推移確率行列を隣接行列として求める

hiragn

2024 年 12 月 12 日

「算数にチャレンジ!!」第 1301 回の問題[*1]を解いた。

> ボタンを押すとランプが赤または緑に光る箱 A, B がある。
>
> - 箱 A は確率 2/3 で赤く光り，確率 1/3 で緑に光る
> - 箱 B は確率 1/3 で赤く光り，確率 2/3 で緑に光る
>
> 1 回目は箱 A を選んでボタンを押す。2 回目以降は前の回に赤く光ったなら次も同じ箱を，緑に光ったなら異なる箱を選んでボタンを押す。
> 10 回目にボタンを押したときランプが赤く光る確率を求めよ。
>
> `http://www.sansu.org/used-html/index1301.html`

これはマルコフ過程。推移確率行列を作って 10 乗を計算すれば解ける。

「推移確率行列の成分をチマチマ指定するかわりに重みつきグラフを作って，その隣接行列を求める方が楽そう」と思ってやってみたらうまくいった。

また，普通に漸化式を使って解くこともできた。

1. 推移確率行列を求める

(a, red), $(a, green)$, (b, red), $(b, green)$ を頂点とするグラフを作る。辺の重みは推移確率で，たとえば箱 A が赤く光ったら次も A を選ぶので次のようになる。

- $(a, red) \to (a, red)$ の確率は単に A が赤く光る確率で 2/3
- $(a, red) \to (a, green)$ の確率は単に A が緑に光る確率で 1/3
- $(a, red) \to (b, red)$ と $(a, red) \to (b, green)$ の確率はともに 0。辺は張らない

[*1] この問題文は原題を適当に書き換えたもの。

他の辺についても同様に考えてグラフを作ってその隣接行列 M を求める。

```
g = Graph[{
    {a, red} -> {a, red}, {a, red} -> {a, green},
    {a, green} -> {b, red}, {a, green} -> {b, green},
    {b, red} -> {b, red}, {b, red} -> {b, green},
    {b, green} -> {a, red}, {b, green} -> {a, green}},
   EdgeWeight -> {p, q, q, p, q, p, p, q},
   VertexLabels -> Automatic, EdgeLabels -> "EdgeWeight"];
m = WeightedAdjacencyMatrix@g;
MatrixForm@m
```

$p = 2/3$, $q = 1/3$ として M は次のようになる。

$$M = \begin{pmatrix} p & q & 0 & 0 \\ 0 & 0 & q & p \\ 0 & 0 & q & p \\ p & q & 0 & 0 \end{pmatrix}$$

- 1 回目は A を選ぶので確率 p で (a, red) になり，確率 q で $(a, green)$ になる。これらは第 1 行，第 2 行に対応する
- ランプが赤く光るのは (a, red) か (b, red) のときで，これらは第 1 列，第 3 列に対応する

行列 X の (i, j) 成分を $X[i, j]$ で表すと，答えは次の式でその値は 5/9。

$$p(M^{10}[1,1] + M^{10}[1,3]) + q(M^{10}[2,1] + M^{10}[2,3])$$

```
In[]:= Clear["Global`*"];
RepeatedTiming[
 {p, q} = {2/3, 1 - p};
 g = Graph[{
    {a, red} -> {a, red}, {a, red} -> {a, green},
    {a, green} -> {b, red}, {a, green} -> {b, green},
    {b, red} -> {b, red}, {b, red} -> {b, green},
    {b, green} -> {a, red}, {b, green} -> {a, green}},
   EdgeWeight -> {p, q, q, p, q, p, p, q},
   VertexLabels -> Automatic, EdgeLabels -> "EdgeWeight"];
 m = WeightedAdjacencyMatrix@g;
 ans = p (#[[1, 1]] + #[[1, 3]]) + q (#[[2, 1]] + #[[2, 3]]) &@
   MatrixPower[m, 10]]

Out[]= {0.0000532011, 5/9}
```

2.　漸化式を解く

次は普通に漸化式を使って解く。

まともにやると 4 つの状態に対応する 4 つの項が必要になるので「どの箱を選ぶか」「何色に光るか」に分けて考える。

n 回目に A, B を選ぶ確率をそれぞれ A_n, B_n とします。$A_n + B_n = 1$ は明らか。

$n+1$ 回目に A を選ぶのは「n 回目が A で赤」または「n 回目が B で緑」のとき。

$$A_{n+1} = pA_n + pB_n = p(A_n + B_n) = p$$

$n+1$ 回目に B を選ぶのは A を選ばないとき。

$$B_{n+1} = 1 - A_{n+1} = 1 - p = q$$

1 回目だけは $A_1 = 1$, $B_1 = 0$ だけれども，その後はずっと $A_n = p$, $B_n = q$ だとわかる。

A, B が赤く光る確率はそれぞれ p, q なので求める確率は次のようになる。

$$A_{10} \cdot p + B_{10} \cdot q = p^2 + q^2 = \frac{5}{9}$$

色を一旦無視することに気づければ，こちらの方がずっと簡単。