「A は連続していいが，B と C は連続しない」順列の個数を再帰で求める

hiragn

2024 年 12 月 12 日

「算数にチャレンジ!!」第 1286 回の問題[*1]を再帰などの方法で解いた。

> A, B, C と書かれたカードがそれぞれ 3, 3, 2 枚の合計 8 枚ある。これらを横一列に並べるとき，B も C も連続しない方法は何通りあるか。
>
> `http://www.sansu.org/used-html/index1286.html`

1.　漸化式を立てて再帰で解く

大学入試によくある「白は連続してもいいが，黒は連続しない順列」の個数を求める問題と同様に考えると，漸化式を立てることができる。

最後の 1 枚が A, B, C のどれなのかに注目する。
「A, B, C をそれぞれ i, j, k 枚並べた順列で，最後が t（0, 1, 2 が A, B, C に対応）のものの個数」を $dp(i, j, k, t)$ であらわす。
添字の範囲を一旦無視すると，漸化式はこうなる。

$$\begin{aligned}
dp(i, j, k, 0) &= dp(i-1, j, k, 0) + dp(i-1, j, k, 1) + dp(i-1, j, k, 2) \\
dp(i, j, k, 1) &= dp(i, j-1, k, 0) + dp(i, j-1, k, 2) \\
dp(i, j, k, 2) &= dp(i, j, k-1, 0) + dp(i, j, k-1, 1)
\end{aligned}$$

初期条件その他はこう。

$$\begin{aligned}
&dp(1, 0, 0, 0) = 1,\ dp(0, 1, 0, 1) = 1,\ dp(0, 0, 1, 2) = 1 \\
&dp(i, j, k, t) = 0 \quad (i, j, k \text{ の少なくとも 1 つが負})
\end{aligned}$$

この漸化式を使ってメモ化再帰で解くと，答えは 160 通りであることがわかる。

[*1] この問題文は原題を適当に書き換えたもの。

```
In[]:= Clear["Global`*"];
RepeatedTiming[
 dp[1, 0, 0, 0] = 1;
 dp[0, 1, 0, 1] = 1;
 dp[0, 0, 1, 2] = 1;
 dp[i_, j_, k_, t_] /; (i < 0 || j < 0 || k < 0) := dp[i, j, k, t] = 0;
 dp[i_, j_, k_, 0] :=
  dp[i, j, k, 0] = Sum[dp[i - 1, j, k, t], {t, 0, 2}];
 dp[i_, j_, k_, 1] :=
  dp[i, j, k, 1] = dp[i, j - 1, k, 0] + dp[i, j - 1, k, 2];
 dp[i_, j_, k_, 2] :=
  dp[i, j, k, 2] = dp[i, j, k - 1, 0] + dp[i, j, k - 1, 1];
 ans = Sum[dp[3, 3, 2, t], {t, 0, 2}]]

Out[]= {0.0000126398, 160}
```

2.　グラフの辺を数える

カードの置き場所を左から順に 1, 2, $\cdots$, 8 とします。A, B, C を並べることはその置き場所を決めることと同じ。

条件をみたす置き方は次のようにして決めることができる。

1. B の置き場所として階差が 1 より大きい 3 つの数を選ぶ
2. 残りの 5 数の中から C の置き場所として階差が 1 より大きい 2 つの数を選ぶ
3. 残った 3 数は A の置き場所で，これは自動的に決まる

1〜8 の中から階差が 1 より大きい n 個の数を選ぶ関数 $g(n)$ を書いて，B の置き場所 $g(3)$ を求めると 20 通りあった。

```
In[]:= g[n_] := Select[Subsets[Range@8, {n}], Min@Differences@# > 1 &];
g@3

Out[]= {{1, 3, 5}, {1, 3, 6}, {1, 3, 7}, {1, 3, 8}, {1, 4, 6},
{1, 4, 7}, {1, 4, 8}, {1, 5, 7}, {1, 5, 8}, {1, 6, 8}, {2, 4, 6},
{2, 4, 7}, {2, 4, 8}, {2, 5, 7}, {2, 5, 8}, {2, 6, 8}, {3, 5, 7},
{3, 5, 8}, {3, 6, 8}, {4, 6, 8}}
```

C の置き場所については手順 2 のように「残りの 5 数の中から〜」とするかわりに $g(2)$ として求めて，$g(3)$ と共通要素がないものを選ぶことにした。

RelationGraph を使って共通要素がない $g(3)$ と $g(2)$ の元の間に辺を張る。その本数が答え。漸化式を使う解法に比べるとかなり遅い。

```
In[]:= Clear["Global‘*"];
RepeatedTiming[
 g[n_] := Select[Subsets[Range@8, {n}], Min@Differences@# > 1 &];
 ans = EdgeCount@RelationGraph[DisjointQ, g@2, g@3]]

Out[]= {0.00815325, 160}
```

ちなみに $g(2)$ の元は 21 個だった。

3.　愚直に順列を数える

ためしに $\{0,0,0,1,1,1,2,2\}$ の順列を全部作って条件をみたすものを数えるコードも書いてみた。順列は全部で

$$\frac{8!}{3!\,3!\,2!} = 560 \text{ 個}$$

SequenceCount を使って 1 も 2 も連続しないものを数える。

```
In[]:= Clear["Global‘*"];
AbsoluteTiming[
 lst = Permutations[{0, 0, 0, 1, 1, 1, 2, 2}];
 cond[lst_] := SequenceCount[lst, {1, 1} | {2, 2}] == 0;
 ans = Length@Select[lst, cond]]

Out[]= {0.031509, 160}
```

漸化式を使う解法の約 2500 倍の時間がかかるが，書くのはこれが圧倒的に楽。